Phase Compensation in a Collinear Source of Polarization-Entangled Photon Pairs at Nondegenerate Wavelengths

Contents

1 Introduction

Photon pairs can be generated in nonlinear crystals by spontaneous parametric down-conversion (SPDC). A source of polarization entangled photon pairs can be generated by using [4]:



Figure 1: Collinear photon pair source (picture taken from [3]).

- type-I SPDC: pump beam is extraordinary polarized, signal and idler photons ordinary
$$e \rightarrow o + o \tag{1}$$
- two identical BBO crystals, type-I phase matching with their optic axis lying in the vertical and horizontal planes
- pump beam polarization orientated under 45°
- horizontally polarized photon pairs $|H\rangle_s|H\rangle_i = |HH\rangle$ are generated in the first and vertically polarized photons pairs $|V\rangle_s|V\rangle_i = |VV\rangle$ in the second crystal.
- The phase matching conditions
$$\begin{aligned} \omega_p &= \omega_s + \omega_i \\ \vec{k}_p &= \vec{k}_s + \vec{k}_i \end{aligned} \tag{2}$$
are fulfilled. $\vec{k}_s$ and $\vec{k}_i$ are the wave vectors. ω_s and ω_i the angular frequencies of the so called signal and idler photons.
- The BBO crystals are cut for collinear type I SPDC with $\lambda_s \neq \lambda_i$. The photons are separated by a dispersive element.
- BBO and YVO4 crystals are used to compensate walk-off effects.

Polarization entanglement can only be observed if the phase between the $|HH\rangle$ and $|VV\rangle$ state is independent of the angular frequencies ω_p of the pump beam and the signal ω_s and idler ω_i photons. The quantum state at single frequencies of the pump beam and a single frequency ω_s (and ω_i) can be written as

$$|\phi(\omega_p,\omega_s,\omega_i)\rangle = |HH\rangle + e^{i\varphi(\omega_p,\omega_s,\omega_i)}|VV\rangle. \tag{3}$$

In general the signal and idler photons have a broad spectrum and the phase φ is a function of ω_p, ω_s and ω_i. The quantum state of a photon pair is superposition of all angular frequencies

$$|\phi\rangle = \iiint \mathrm{d}\omega_p \, \mathrm{d}\omega_s \, \mathrm{d}\omega_i \, A(\omega_p,\omega_s,\omega_i) \left(|HH\rangle + e^{i\varphi(\omega_p,\omega_s,\omega_i)}|VV\rangle\right) |\omega_s\omega_i\rangle. \qquad (4)$$

If the phase φ is independent of ω_p, ω_s and ω_i the quantum state can be written as a product of the angular frequency and polarization state

$$\begin{aligned} |\phi\rangle &= \left(|HH\rangle + e^{i\varphi}|VV\rangle\right) \otimes \left(\iiint \mathrm{d}\omega_p \, \mathrm{d}\omega_s \, \mathrm{d}\omega_i \, A(\omega_p,\omega_s,\omega_i)|\omega_s\omega_i\rangle\right) \\ &= |\phi\rangle \otimes |\phi_\omega\rangle \end{aligned} \qquad (5)$$

and the polarization state

$$|\phi\rangle = |HH\rangle + e^{i\varphi}|VV\rangle \qquad (6)$$

is maximally entangled. To observe a polarization entangled state it is therefor necessary that the phase φ is independent of ω_p, ω_s and ω_i. The phase $\varphi(\omega_p,\omega_s,\omega_i)$ is calculated in the following chapters. In all calculations a collinear plane wave model will be used. Transverse walk-off effects can be compensated with a additional pair of BBO crystals. The optic axes of these two crystals are crossed and therefore they have no influence on the longitudinal walk-off [3].
If the phase depends on the angular frequencies the polarization state can be calculated using the density matrix formalism (see chapter 5).



Figure 2: Experimental setup of a collinear entangled photon pair source [3].

2 Polarization and Propagation in Free Space

A plane wave with the polarization

$$|P\rangle = \sin\alpha|H\rangle + e^{i\varphi}\cos\alpha|V\rangle, \tag{7}$$

the wave vector $\vec{k}$ and the angular frequency ω can be described by the state vector

$$|\psi_{\vec{k},\omega}\rangle = |\vec{k},\omega,P\rangle = |\vec{k}\rangle \otimes |\omega\rangle \otimes |P\rangle. \tag{8}$$

The phase of the plane wave changes with position $\vec{r}$ and time t

$$|\psi_{\vec{k},\omega}(\vec{r},t)\rangle = e^{i(\vec{k}\cdot\vec{r}-\omega t)}|\vec{k},\omega,P\rangle. \tag{9}$$

The quantum state of a single photon can be described by a superposition of plane plane waves

$$|\psi(\vec{r},t)\rangle = \int_0^\infty \mathrm{d}\omega \iiint_{-\infty}^{\infty} \mathrm{d}^3\vec{k}\, A(\omega,\vec{k})\, e^{i(\vec{k}\cdot\vec{r}-\omega t)}\, |\vec{k},\omega,P\rangle. \tag{10}$$

$A(\omega,\vec{k})$ is a complex function with $\int_0^\infty \mathrm{d}\omega \iiint_{-\infty}^{\infty} \mathrm{d}^3\vec{k}\, |A(\omega,\vec{k})|^2 = 1$. The polarization does not change during propagation. The polarization part of the state is separable

$$|\psi(\vec{r},t)\rangle = \left(\int_0^\infty \mathrm{d}\omega \iiint_{-\infty}^{\infty} \mathrm{d}^3\vec{k}\, A(\omega,\vec{k})\, e^{i(\vec{k}\cdot\vec{r}-\omega t)}\, |\vec{k},\omega\rangle \right) \otimes |P\rangle. \tag{11}$$

3 Polarization and Propagation in Birefringent Crystals

In birefringent crystals there are two orthogonal polarization eigenstates. The ordinary polarization $|o\rangle$ and the extraordinary polarization $|e\rangle$. Birefringent crystals can be oriented in a way that $|H\rangle = |o\rangle$ and $|V\rangle = |e\rangle$ (or $|H\rangle = |e\rangle$ and $|V\rangle = |o\rangle$). The extraordinary refractive index $n_e(\omega,\theta)$ is a function of the angular frequency ω and the angle θ of the crystal to the optic axis. The ordinary refractive index $n_o(\omega)$ depends only on the angular frequency. The extraordinary refractive index is [2]

$$\frac{1}{[n_e(\theta)]^2} = \frac{\cos^2(\theta)}{n_o^2} + \frac{\sin^2(\theta)}{n_e^2} \tag{12}$$

with the principal refractive indices $n_x = n_y = n_o$ and $n_z = n_e$.
The wave vectors inside a birefrigent crystal are [1]

$$\begin{aligned} \vec{k}_o &= \frac{2\pi n_o(\lambda_{\mathrm{vac}})}{\lambda_{\mathrm{vac}}} \cdot \vec{e}_k = n_o(\omega) \cdot \frac{\omega}{c} \cdot \vec{e}_k \\ \vec{k}_e &= \frac{2\pi n_e(\lambda_{\mathrm{vac}},\theta)}{\lambda_{\mathrm{vac}}} \cdot \vec{e}_k = n_e(\omega,\theta) \cdot \frac{\omega}{c} \cdot \vec{e}_k \end{aligned} \tag{13}$$

where the dispersion relation of plane waves in vacuum

$$\omega = ck_{\mathrm{vac}} = c\frac{2\pi}{\lambda_{\mathrm{vac}}} \tag{14}$$

has been used.
A plane wave with polarization

$$|P\rangle = \sin\alpha|o\rangle + e^{i\varphi}\cos\alpha|e\rangle \tag{15}$$

has to be discribed using the two wave vectors $\vec{k}_o$ and $\vec{k}_e$ and the angular frequency ω. The state vector of the plane wave is

$$\begin{aligned} |\psi_{\vec{k},\omega}(\vec{r},t)\rangle &= e^{i(\vec{k}_o\cdot\vec{r}-\omega t)}\sin\alpha|\vec{k}_o,\omega,o\rangle + e^{i(\vec{k}_e\cdot\vec{r}-\omega t)+i\varphi}\cos\alpha|\vec{k}_e,\omega,e\rangle \\ &= e^{i(\vec{k}_o\cdot\vec{r}-\omega t)}\left(\sin\alpha|\vec{k}_o,\omega,o\rangle + e^{i(\vec{k}_e-\vec{k}_o)\cdot\vec{r}+i\varphi}\cos\alpha|\vec{k}_e,\omega,e\rangle\right). \end{aligned} \tag{16}$$

The phase between the ordinary and extraordinary polarization changes with the position $\vec{r}$. Therefore the polarization inside a crystal changes if it is not in one of the two eigenstates $|o\rangle$ or $|e\rangle$.
The quantum state of a single photon can be described by a superposition of plane plane waves

$$\begin{aligned} |\psi(\vec{r},t)\rangle = \int_0^\infty \mathrm{d}\omega \iiint_{-\infty}^{\infty} \mathrm{d}^3\vec{k}_o \iiint_{-\infty}^{\infty} \mathrm{d}^3\vec{k}_e \\ \left(A(\omega,\vec{k}_o,\vec{k}_e)\, e^{i(\vec{k}_o\cdot\vec{r}-\omega t)}(\sin\alpha|\vec{k}_o,\omega,o\rangle + e^{i(\vec{k}_e-\vec{k}_o)\cdot\vec{r}+i\varphi}\cos\alpha|\vec{k}_e,\omega,e\rangle)\right) \end{aligned} \tag{17}$$

where $A(\omega,\vec{k})$ is a complex function with $\int_0^\infty \mathrm{d}\omega \iiint_{-\infty}^{\infty} \mathrm{d}^3\vec{k}_o \iiint_{-\infty}^{\infty} \mathrm{d}^3\vec{k}_e\, |A(\omega,\vec{k}_o,\vec{k}_e)|^2 = 1$.The polarization degree of freedom is not separable anymore. In general the polarization is not in a pure but in a mixed state.

3.1 Plane Waves in z Direction

In a collinear SPDC source only plane waves propagating in z-direction have to be considered. The ordinary wave vector k_o and the extraordinary wave vector k_e in z-direction depend on the angular frequency ω

$$\begin{aligned} k_o(\omega) &= n_o(\omega) \cdot \frac{\omega}{c} \\ k_e(\omega) &= n_e(\omega, \theta) \cdot \frac{\omega}{c} \end{aligned} \tag{18}$$

and the state vector of a single photon with polarization $|P\rangle$ (eqn. 17) reduces to [1]

$$|\psi(\vec{r},t)\rangle = \int_0^\infty \mathrm{d}\omega\, A(\omega)\, e^{i(k_o(\omega)\cdot z - \omega t)} \left(\sin\alpha|\omega, o\rangle + e^{i(k_e(\omega)-k_o(\omega))\cdot z + i\varphi} \cos\alpha|\omega, e\rangle\right) \tag{19}$$

The phase ϕ between the ordinary and extraordinary polarization behind a crystal of the Length L is

$$\phi = (k_e(\omega, \theta) - k_o(\omega)) \cdot L \tag{20}$$

and depends on the angular frequency ω. $|\psi(\vec{r},t)\rangle$ is in general not separable in polarization and energy. The polarization state can be calculated by the partial trace over ω

$$\rho_p = \mathrm{Tr}_\omega\{\rho\} = \mathrm{Tr}_\omega\{|\psi\rangle\langle\psi|\} \tag{21}$$

3.2 Photon Pair Propagation in z Direction



Figure 3: Birefringent crystal of length L. The polarization state before the crystal is $|HH\rangle + e^{i\varphi}|VV\rangle$. The extraordinary polarization can be oriented in the horizontal or vertical plane.

[1] k_o only depends on the angular frequency. k_e on the angular frequency and the (fixed) cut angle of the optic axis to the propagation direction. Therefore the simpler notation $|\omega, o\rangle = |k_o(\omega), \omega, o\rangle$ and $|\omega, e\rangle = |k_e(\omega), \omega, e\rangle$ can be used.

Any pure polarization state of a photon pair can be written as

$$\begin{aligned}|P_s P_i\rangle &= a_{oo}|o\rangle|o\rangle + a_{ee}|e\rangle|e\rangle + a_{oe}|o\rangle|e\rangle + a_{eo}|e\rangle|o\rangle \\ &= \sum_{s=o,e}\sum_{i=o,e} a_{si}|s\rangle|i\rangle.\end{aligned} \tag{22}$$

To keep the notation simpler the calculations will only be made for the polarization states produced by type I SPDC

$$|P_s P_i\rangle = \sin\alpha|o\rangle|o\rangle + e^{i\varphi}\cos\alpha|e\rangle|e\rangle \tag{23}$$

The photon pair plane wave state is

$$\begin{aligned}|\psi_{\omega_s,\omega_i}(\vec{r},t)\rangle &= \sin\alpha\left(e^{i(k_o(\omega_s)\cdot z-\omega_s t)}|\omega_s,o\rangle_s e^{i(k_o(\omega_i)\cdot z-\omega_i t)}|\omega_i,o\rangle_i\right) \\ &\quad + e^{i\varphi}\cos\alpha\left(e^{i(k_e(\omega_s)\cdot z-\omega_s t)}|\omega_s,e\rangle_s e^{i(k_e(\omega_i)\cdot z-\omega_i t)}|\omega_i,e\rangle_i\right) \\ &= e^{-i(\omega_s+\omega_i)t}e^{i(k_o^s+k_o^i)\cdot z} \\ &\quad\cdot\Big(\sin\alpha|\omega_s,o\rangle_s|\omega_i,o\rangle_i + e^{i\varphi}\cos\alpha e^{i(k_e(\omega_s)-k_o(\omega_s)+k_e(\omega_i)-k_o(\omega_i))\cdot z}|\omega_s,e\rangle_s|\omega_i,e\rangle_i\Big).\end{aligned} \tag{24}$$

The photon pair state is a superposition of these plane waves

$$\begin{aligned}|\psi(\vec{r},t)\rangle &= \int_0^\infty \mathrm{d}\omega_s \int_0^\infty \mathrm{d}\omega_i\, A(\omega_s,\omega_i)\, e^{-i(\omega_s+\omega_i)t}e^{i(k_o^s+k_o^i)\cdot z} \\ &\quad\cdot\Big(\sin\alpha|\omega_s,o\rangle_s|\omega_i,o\rangle_i + e^{i\varphi}\cos\alpha\, e^{i(k_e(\omega_s)-k_o(\omega_s)+k_e(\omega_i)-k_o(\omega_i))\cdot z}|\omega_s,e\rangle_s|\omega_i,e\rangle_i\Big).\end{aligned} \tag{25}$$

The phase between $|o\rangle|o\rangle$ and $|e\rangle|e\rangle$

$$\phi_2(\omega_s,\omega_i) = (k_e(\omega_s) - k_o(\omega_s) + k_e(\omega_i) - k_o(\omega_i))\cdot z \tag{26}$$

is the sum of the two phases and depends on the angular frequencies ω_s and ω_i. Therefore the polarization state is in general a mixed state. It can be calculated by the partial trace over ω_s and ω_i

$$\rho_p = \mathrm{Tr}_{\omega_s,\omega_i}\{\rho\} = \mathrm{Tr}_{\omega_s,\omega_i}\{|\psi\rangle\langle\psi|\}. \tag{27}$$

4 Phases in Collinear SPDC Sources

The phases in a type I SPDC two crystal source will be calculated in three steps. First the phases if only two BBO crystals are used. Then the post-compensation crystal behind the BBO crystals is added. And last the pre-compensation crystal in the pump beam is added.



Figure 4: Pair of BBO crystals cut for type I SPDC. Difference between $|HH\rangle$ and $|VV\rangle$ photons: signal and idler photons generated in the first crystal have to pass the second crystal. The ordinary polarized part of the pump beam has to pass the first crystal before photon pairs can be generated in the second crystal.

4.1 BBO SPDC Crystals

The horizontal direction H and vertical direction V can be chosen arbitrarily. In the following calculations the orientation of the two crystals is

- **SPDC in first crystal:**
  - V is extraordinary and H is ordinary polarization.
  - V part of the pump beam generates horizontally polarized photon pairs
    $$|HH\rangle = |\omega_s, H\rangle_s |\omega_i, H\rangle_i = |\omega_s, \omega_i\rangle_{s,i} |H, H\rangle_{s,i} \tag{28}$$
  - $|HH\rangle$ photon pairs have to pass the second BBO crystal. H is extraordinary polarization in the second crystal.
- **SPDC in second crystal:**
  - H is extraordinary and V is ordinary polarization.
  - H part of the pump beam generates vertically polarized photon pairs
    $$|VV\rangle = |\omega_s, V\rangle_s |\omega_i, V\rangle_i = |\omega_s, \omega_i\rangle_{s,i} |V, V\rangle_{s,i} \tag{29}$$
  - The H part of the pump beam has to pass the first BBO crystal. H is ordinary polarization in the first crystal.

The difference between the $|HH\rangle$ photons generated in the first BBO crystal and the $|VV\rangle$ photons generated in the second crystal is:

- $|HH\rangle$ photons have to pass the second crystal.
- The H polarized part of the pump beam has to pass the first BBO crystal before $|VV\rangle$ photon pairs can be generated in the second crystal.

Only these differences have to be considered to calculate the phase of the $|HH\rangle$ and $|VV\rangle$ photons.

Phase of $|HH\rangle$ Photons

The $|HH\rangle$ photons generated in the first crystal are extraordinary polarized in the second BBO crystal

$$|\omega_s, e\rangle|\omega_i, e\rangle. \tag{30}$$

In the second crystal these down conversion photons are extraordinary polarized and the phase ϕ_{BBO}^{HH} of the $|HH\rangle$ photons behind the second BBO crystal of length L_{BBO} is (eqn. 24)

$$e^{ik_e(\omega_s)L_{\mathrm{BBO}}+ik_e(\omega_i)L_{\mathrm{BBO}}}|\omega_s, H\rangle|\omega_i, H\rangle = e^{i\phi_{\mathrm{BBO}}^{HH}}|\omega_s, H\rangle|\omega_i, H\rangle. \tag{31}$$

$k_e(\omega_s)$ and $k_e(\omega_i)$ are the wave vectors in BBO for extraordinary polarization.

Phase of $|VV\rangle$ Photons

The H polarized part of the pump beam passes the first crystal without generating photon pairs. The phase ϕ_{BBO}^{VV} of the $|VV\rangle$ of the pump beam behind the first crystal is

$$e^{ik_o(\omega_p)L_{\mathrm{BBO}}}|\omega_p, o\rangle \tag{32}$$

Due to the coherence in SPDC the phase of the $|VV\rangle$ photons is

$$e^{ik_o(\omega_p)L_{\mathrm{BBO}}}|\omega_s, V\rangle|\omega_i, V\rangle = e^{i\phi_{\mathrm{BBO}}^{VV}}|\omega_s, V\rangle|\omega_i, V\rangle. \tag{33}$$

Phase between $|HH\rangle$ and $|VV\rangle$

The quantum state $|\psi_{\omega_p,\omega_s,\omega_i}\rangle$ at the angular frequencies ω_p of the pump beam, ω_s of the signal and ω_i of the idler photon is

$$|\psi_{\omega_p,\omega_s,\omega_i}\rangle = e^{i\phi_{\mathrm{BBO}}^{HH}}|\omega_s, H\rangle|\omega_i, H\rangle + e^{i\phi_{\mathrm{BBO}}^{VV}}|\omega_s, V\rangle|\omega_i, V\rangle \tag{34}$$

The photon pair state is a superposition of the states $|\psi_{\omega_p,\omega_s,\omega_i}\rangle$

$$|\psi_{\mathrm{BBO}}\rangle = \iiint_0^\infty \mathrm{d}\omega_p\,\mathrm{d}\omega_s\,\mathrm{d}\omega_i\, A(\omega_p, \omega_s, \omega_i)\,|\psi_{\omega_p,\omega_s,\omega_i}\rangle \tag{35}$$

The phase between $|HH\rangle$ and $|VV\rangle$

$$\phi_{\mathrm{BBO}} = \phi_{\mathrm{BBO}}^{VV} - \phi_{\mathrm{BBO}}^{HH} = (k_o(\omega_p) - k_e(\omega_s) - k_e(\omega_i))\,L_{\mathrm{BBO}} \tag{36}$$

depends on the angular frequencies ω_p, ω_s and ω_i. Therefore the broadband state $|\psi_{\mathrm{BBO}}\rangle$ is not entangled in the polarization degree of freedom. Additional crystals are needed to compensate the dependence of the phase on ω_p, ω_s and ω_i.



Figure 5: Compensation crystal of length L_c behind a pair of BBO crystals. The additional *ee* path of the photons generated in the first BBO crystal is compensated.

4.2 Post Compensation Crystal

A crystal behind the two SPDC BBO crystals can be used to compensate the phase dependence on ω_s and ω_i. One possible crystal is YVO4 with the orientation shown in fig. 5. The wave vectors inside the crystal are

$$\begin{aligned} k_o^{\mathrm{YVO4}}(\omega) &= n_o^{\mathrm{YVO4}}(\omega) \cdot \frac{\omega}{c} \\ k_e^{\mathrm{YVO4}}(\omega) &= n_e^{\mathrm{YVO4}}(\omega, \theta) \cdot \frac{\omega}{c} \end{aligned} \tag{37}$$

and the additional phase from the post compensation crystal is

$$\begin{aligned} |\omega_s, H\rangle|\omega_i, H\rangle : &\quad e^{ik_o^{\mathrm{YVO4}}(\omega_s)L_c + ik_o^{\mathrm{YVO4}}(\omega_i)L_c} = e^{i\phi_c^{HH}} \\ |\omega_s, V\rangle|\omega_i, V\rangle : &\quad e^{ik_e^{\mathrm{YVO4}}(\omega_s)L_c + ik_e^{\mathrm{YVO4}}(\omega_i)L_c} = e^{i\phi_c^{VV}} \end{aligned} \tag{38}$$

The quantum state $|\psi^{\mathrm{c}}_{\omega_p,\omega_s,\omega_i}\rangle$ behind the post compensation crystal at the angular frequencies ω_p of the pump beam, ω_s of the signal and ω_i of the idler photon is

$$|\psi^{\mathrm{c}}_{\omega_p,\omega_s,\omega_i}\rangle = e^{i\phi_{\mathrm{BBO}}^{HH} + i\phi_c^{HH}}|\omega_s, H\rangle|\omega_i, H\rangle + e^{i\phi_{\mathrm{BBO}}^{VV} + i\phi_c^{VV}}|\omega_s, V\rangle|\omega_i, V\rangle. \tag{39}$$

The post compensation crystal phases ϕ_c^{HH} and ϕ_c^{VV} only depend on ω_s and ω_i. If the pump beam has a broad bandwidth an additional crystal is needed.

4.3 Pre-Comepensation Crystal

A crystal in front of two SPDC BBO crystals can be used to compensate the phase dependence on ω_p. One possible crystal is YVO4 with the orientation shown in fig. 6. The additional phase from the pre-compensation crystal is

$$\begin{aligned} \text{pump beam } V : &\quad e^{ik_e^{\mathrm{YVO4}}(\omega_p)L_{\mathrm{pc}}} \\ \text{pump beam } H : &\quad e^{ik_o^{\mathrm{YVO4}}(\omega_p)L_{\mathrm{pc}}} \end{aligned} \tag{40}$$

Due to the coherence in SPDC the phases induced by the pre-compensation crystal are

$$\begin{aligned} \text{signal, idler } |\omega_s, H\rangle|\omega_i, H\rangle : &\quad e^{ik_e^{\mathrm{YVO4}}(\omega_p)L_{\mathrm{pc}}} = e^{i\phi_{\mathrm{pc}}^{HH}} \\ \text{signal, idler } |\omega_s, V\rangle|\omega_i, V\rangle : &\quad e^{ik_o^{\mathrm{YVO4}}(\omega_p)L_{\mathrm{pc}}} = e^{i\phi_{\mathrm{pc}}^{VV}} \end{aligned} \tag{41}$$



Figure 6: Pre-compensation crystal of length L_{pc} before a pair of BBO crystals. The additional o path of the pump beam in the first BBO crystal is compensated.

The phases ϕ_{pc}^{HH} and ϕ_{pc}^{VV} induced by the pre-compensation crystal only depend on the angular frequency ω_p of the pump beam.

4.4 Complete Phase Behind all Four Crystals

The quantum state $|\psi_{\omega_p,\omega_s,\omega_i}^{\text{pc,BBO,c}}\rangle$ behind all four crystals at the angular frequencies ω_p of the pump beam, ω_s of the signal and ω_i of the idler photon is

$$\begin{aligned}|\psi_{\omega_p\omega_s\omega_i}^{\text{pc,BBO,c}}\rangle &= e^{i\phi_{\text{BBO}}^{HH}+i\phi_{\text{c}}^{HH}+i\phi_{\text{pc}}^{HH}}|\omega_s,H\rangle|\omega_i,H\rangle + e^{i\phi_{\text{BBO}}^{VV}+i\phi_{\text{c}}^{VV}+i\phi_{\text{pc}}^{VV}}|\omega_s,V\rangle|\omega_i,V\rangle|\omega_s\omega_i\rangle \\ &= e^{i\phi_{\text{BBO}}^{HH}+i\phi_{\text{c}}^{HH}+i\phi_{\text{pc}}^{HH}}\left(|HH\rangle + e^{i(\phi_{\text{BBO}}^{VV}-\phi_{\text{BBO}}^{HH})+i(\phi_{\text{c}}^{VV}-\phi_{\text{c}}^{HH})+i(\phi_{\text{pc}}^{VV}-\phi_{\text{pc}}^{HH})}|VV\rangle\right)\otimes|\omega_s\omega_i\rangle \\ &= e^{i\phi_{\text{BBO}}^{HH}+i\phi_{\text{c}}^{HH}+i\phi_{\text{pc}}^{HH}}\left(|HH\rangle + e^{i\phi(\omega_p,\omega_s,\omega_i)}|VV\rangle\right)\otimes|\omega_s\omega_i\rangle\end{aligned} \tag{42}$$

with the phase

$$\begin{aligned}\phi(\omega_p,\omega_s,\omega_i) &= (\phi_{\text{BBO}}^{VV}-\phi_{\text{BBO}}^{HH}) + (\phi_{\text{c}}^{VV}-\phi_{\text{c}}^{HH}) + (\phi_{\text{pc}}^{VV}-\phi_{\text{pc}}^{HH}) \\ &= (k_e(\omega_s)+k_e(\omega_i)-k_o(\omega_p))\,L_{\text{BBO}} \\ &\quad+\left(k_e^{\text{YVO4}}(\omega_s)+k_e^{\text{YVO4}}(\omega_i)-k_o^{\text{YVO4}}(\omega_s)-k_o^{\text{YVO4}}(\omega_i)\right)L_{\text{c}} \\ &\quad+\left(k_o^{\text{YVO}}(\omega_p)-k_e^{\text{YVO}}(\omega_p)\right)L_{\text{pc}} \\ &= -k_o(\omega_p)L_{\text{BBO}}+\left(k_o^{\text{YVO}}(\omega_p)-k_e^{\text{YVO}}(\omega_p)\right)L_{\text{pc}} \\ &\quad+(k_e(\omega_s)+k_e(\omega_i))L_{\text{BBO}} \\ &\quad+\left(k_e^{\text{YVO4}}(\omega_s)+k_e^{\text{YVO4}}(\omega_i)-k_o^{\text{YVO4}}(\omega_s)-k_o^{\text{YVO4}}(\omega_i)\right)L_{\text{c}}\end{aligned} \tag{43}$$

In general the spectra of the pump beam, the signal and the idler photons are broadband and the quantum state is a superposition of the single angular frequency states

$$\begin{aligned}|\psi\rangle &= \iiint_{-\infty}^{\infty} \mathrm{d}\omega_p\,\mathrm{d}\omega_s\,\mathrm{d}\omega_i\,\tilde{A}(\omega_p,\omega_s,\omega_i)|\psi_{\omega_p\omega_s\omega_i}^{\text{pc,BBO,c}}\rangle \\ &= \iiint_{-\infty}^{\infty} \mathrm{d}\omega_p\,\mathrm{d}\omega_s\,\mathrm{d}\omega_i\,A(\omega_p,\omega_s,\omega_i)\left(|HH\rangle + e^{i\phi(\omega_p,\omega_s,\omega_i)}|VV\rangle\right)|\omega_s\omega_i\rangle.\end{aligned} \tag{44}$$

with $A(\omega_p, \omega_s, \omega_i) \equiv e^{i\phi_{\mathrm{BBO}}^{HH} + i\phi_{\mathrm{c}}^{HH} + i\phi_{\mathrm{pc}}^{HH}} \tilde{A}(\omega_p, \omega_s, \omega_i)$. Only if the phase ϕ does not depend on ω_p, ω_s and ω_i the polarization degree of freedom is in a pure and maximally entangled state. Otherwise the polarization state is different for every angular frequency component ω_s, ω_i and ω_p. The pre-compensation and compensation crystal can be used to make the phase ϕ almost independent of ω_p, ω_s and ω_i. A `mathematica` script to calculate the phase ϕ is shown in 4.6.

4.5 Group Velocity Compensation

To observe entanglement the arrival time of a photon pair has to be same for $|HH\rangle$ and $|VV\rangle$ polarization. Therefore the mean group velocity including the group velocity of the pump beam before SPDC has to be the same for $|HH\rangle$ and $|VV\rangle$ polarized photon pairs. The group velocity v_g is defined by the equation

$$v_g \equiv \frac{\partial \omega}{\partial k} \tag{45}$$

If the phase ϕ between $|HH\rangle$ and $|VV\rangle$ does not depend on ω_s, ω_i and ω_p the arrival time is the same for both polarizations.

4.6 Mathematica Phase Calculations

The phase ϕ between $|HH\rangle$ and $|VV\rangle$ in a collinear type I SPDC source can be calculated with the mathematica program `phase-compensation.nb`
First the working directory and options of the `Plot` and `ListPlot` function are set.

set options

```
In[1]:= (* set working directory *)
SetDirectory[FileNameJoin[{$HomeDirectory, "phase_calculations"}]];
(* Plot and ListPlot options *)
SetOptions[Plot, Frame → True, PlotStyle → {Red, Green, Blue, Black}];
SetOptions[ListPlot, Frame → True,
  PlotStyle → {Red, Green, Blue, Black}];
```

The refractive index in birefringent crystals are described by the Sellmeier equations. The wave vector `kvect` is defined defined along the direction `s`.
θ and ϕ are the polar and azimuthal angles. `LT` defines the length change of a crystal with temperature.

refractive index, vectors, ...

In[4]:=
```
(* extraordinary refractive index in uniaxial crystals *)
nth[no_, ne_, θ_] := no * ne / Sqrt[ne^2 * Cos[θ]^2 + no^2 * Sin[θ]^2]
```

In[5]:=
```
(* unit vector
 θ: polar angle ϕ: azimuthal angle *)
vect[θ_, ϕ_] := {Sin[θ] * Cos[ϕ], Sin[θ] * Sin[ϕ], Cos[θ]};
(* wave vector
 λ: wavelength in vacuum, nbr: refractive index, s: direction *)
kvect[λ_, nbr_, s__] := 2 * π * nbr / λ * Normalize[s];
(* wave vector
 λ: wavelength in vacuum, nbr: refractive index,
θ: polar angle ϕ: azimuthal angle *)
kvect[λ_, nbr_, θ_, ϕ_] := kvect[λ, nbr, vect[θ, ϕ]];
```

In[8]:=
```
(* Length of a crystal at Temperature T, L0 at 20°C *)
LT[L0_, α_, T_] := L0 + L0 * (T - 20) * α;
(* YVO4: α= 11.3 , see YVO4_temperature_12082015.nb *)
(* BBO: αa = 4*10^-6/K, αc=36*10^-6/K,
Foctek homepage *)
```

The YVO4 refractive index `nYVO` is modeled using the Sellmeier equations with the coefficients provided by Foctek. The first function calculates the refractive indexes in the principal planes n_o and n_e. The second function returns the refractive index along the direction with an angle α between the optic axis and the wave vector $\vec{k}$.

Sellmeier Equations

YVO4

In[9]:=
```
(*Sellmeier equations of Yttrium Vanadate crystals,
Handbook of nonlinear optical crystals *)
(* http://www.redoptronics.com/Nd-YVO4-crystal.html
    http://www.foctek.net/products/YVO4.htm
   thermal expansion: aa=4.43x10-6/K,ac=11.37x10-6/K
*)
(* wavelength in μm *)
nYVO[lambda_, temp_, axis : (e | o)] := Module[{nr, x},
   nr = If[MatchQ[axis, o],
     Sqrt[(3.77834 + 0.069736 / (lambda^2 - 0.04724) -
         0.0108133* lambda^2)] * (1 + (temp - 20) * 8.5 * 10^-6),
     Sqrt[4.59905 + 0.110534 / (lambda^2 - 0.04813) -
        0.0122676* lambda^2] * (1 + (temp - 20) * 3.0 * 10^-6)];
   Return[nr]];
(*Refractive index of the extraordinary wave as a function
 of the polar angle (α) between optic axis and k vector *)
nYVOeo[λ_, temp_, α_] := nth[nYVO[λ, temp, o], nYVO[λ, temp, e], α];
```

BBO refractive index is calculated with the function `nBBO`:

BBO

```
In[12]:= (* Sellmeier equations, BBO crystal *)
(* wavelength in μm *)
nBBO[lambda_, temp_, axis : (e | o)] := Module[{nq, x},
  nq = If[MatchQ[axis, o],
    2.7359 + 0.01878 / (lambda^2 - 0.01822) - 0.01354 * lambda^2,
    2.3753 + 0.01224 / (lambda^2 - 0.01667) - 0.01516 * lambda^2];
  Sqrt[nq] +
   (InterpolatingPolynomial[
       If[MatchQ[axis, o], {{1.014, -16.64}, {.579, -16.35},
         {.4047, -16.83}}, {{1.014, -9.76}, {.5790, -9.42},
         {.4047, -8.84}}], x] //. x → lambda) * 10^-6 * (temp - 20)]
(*Refractive index of the extraordinary wave as a function
 of the polar angle (α) between optic axis and k vector*)
nBBOeo[λ_, temp_, θ_] := nth[nBBO[λ, temp, o], nBBO[λ, temp, e], θ];
```

First the phase difference between $|HH\rangle$ and $|VV\rangle$ if only two BBO crystals are used

phase difference in BBO crystals

```
In[16]:= Δdc = (kpo - (kse + kie)) LBBO
Out[16]= (-kie + kpo - kse) LBBO
```

The phase caused by the the pre-compensation YVO4 crystal

phase difference in pre-compensation crystal

H: ordinary ,V: extraordinary
exp(i*kpeYVO*Lpc)|HH> + exp(i*kpoYVO*Lpc)*exp(i*Δdc) |VV>

```
In[17]:= Δpc = (kpoYVO - kpeYV04) * Lpc
Out[17]= (-kpeYV04 + kpoYVO) Lpc
```

The phase caused by the compensation crystal

phase difference in compensation crystal

H(signal) and H(idler) ordinary
V(signal) and V(idler) extraordinary
exp(i*(ksoYVO+kioYVO)*Lc)|HH> + exp(i*(kseYVO+kieYVO)*Lc)*exp(i*(Δdc+Δpc)
|VV>

```
In[18]:= Δc = (kseYVO + kieYVO - (ksoYVO + kioYVO)) * Lc
Out[18]= (kieYVO - kioYVO + kseYVO - ksoYVO) Lc
```

The phase differences cause by all four crystals add up:

phase behind all crystals, dc crystal and compensation with different length and material HH> + exp(i*(Δdc+Δpc +Δc)) |VV>

```
In[19]:= Δpc + Δdc + Δc
Out[19]= (-kie + kpo - kse) LBBO +
  (kieYVO - kioYVO + kseYVO - ksoYVO) Lc + (-kpeYV04 + kpoYVO) Lpc
```

The wave vectors depend on the angular frequency. The angular frequency is replaced by the commonly used vacuum wavelength (in μm). Therefore the replacement rules

```
In[20]:= pckvect = {kpeYVO4 → 2 * π * nYVO[λp, Tpc, e] / λp,
            kpoYVO → 2 * π * nYVO[λp, Tpc, o] / λp};
         ckvect = {ksoYVO -> 2 * π * nYVO[λs, Tc, o] / λs,
            kseYVO -> 2 * π * nYVO[λs, Tc, e] / λs,
            kioYVO -> 2 * π * nYVO[λi, Tc, o] / λi,
            kieYVO -> 2 * π * nYVO[λi, Tc, e] / λi};
         dckvect = {kpo → 2 * π * nBBO[λp, TBBO, o] / λp,
            kse -> 2 * π * nBBOeo[λs, TBBO, θp] / λs,
            kie -> 2 * π * nBBOeo[λi, TBBO, θp] / λi};
```

are defined. Using these rules the phase function ϕdc is defined (equation 43).

```
In[23]:= ϕdc[θp_, λp_, λs_, λi_, LBBO_, Lc_, Lpc_, TBBO_, Tc_, Tpc_] :=
           Evaluate[(Δpc + Δdc + Δc /. pckvect /. dckvect /. ckvect)];
         ϕdc[θp_, λp_, λs_, LBBO_, Lc_, Lpc_, TBBO_, Tc_, Tpc_] :=
          ϕdc[θp, λp, λs, (1 / λp - 1 / λs) ^ (-1), LBBO, Lc, Lpc, TBBO, Tc, Tpc]
```

The second ϕdc function uses energy conservation in the SPDC process. $\omega_p = \omega_s + \omega_i$ or in vacuum wavelength $\frac{1}{\lambda_p} = \frac{1}{\lambda_s} + \frac{1}{\lambda_i}$.

4.7 Phase Calculations Results

Source parameters used in the calculations

- BBO length $L_{\mathrm{BBO}} = 6\,\mathrm{mm} = 6000\,\mu\mathrm{m}$
- YVO4 compensation crystal $L_c = 3120\,\mu\mathrm{m}$
- YVO4 pre-compensation crystal $L_{\mathrm{pc}} = 3440\,\mu\mathrm{m}$
- pump wavelength $\lambda_p = 0.406\,\mu\mathrm{m}$, $\lambda_s = 0.765\,\mu\mathrm{m}$, λ_i from energy conservation $\frac{1}{\lambda_p} = \frac{1}{\lambda_s} + \frac{1}{\lambda_i}$

phase compensation calculations

```
In[85]:= sourceparameters = {L → 6000., Lc → 3120., Lpc → 3440., TBBO → 20,
            Tc → 20, Tpc → 20, λp → 0.406`, λs → .765, Tbbo → 20, ϕp → π/2,
            θp → 0.5026584054092709`}
Out[85]= {L → 6000., Lc → 3120., Lpc → 3440., TBBO → 20, Tc → 20, Tpc → 20,
          λp → 0.406, λs → 0.765, Tbbo → 20, ϕp → π/2, θp → 0.502658}
```

4.7.1 Two BBOs, Compensation and Pre-Compensation Crystal

- monochromatic pump beam
- constant temperature

The phase depends on the signal wavelength λ_s:

```
In[86]:= Plot[Mod[180/π (ϕdc[θp, λp, λS, L, Lc, Lpc, TBBO, Tc, Tpc]) /.
            sourceparameters, 360], {λS, .740, .9},
          FrameLabel→{"λ in μm", "phase in degrees"}, ImageSize→400,
          PlotRange→All]
```

Out[86]=



The change over the whole spectrum is smaller than 5 degrees. The range does not change if the pre-compensation crystal is removed (length `Lc` is set to zero).

4.7.2 Compensation Crystal Length

The influence of the compensation crystal length `LC` on the phase compensation is calculated. The phases $\phi(\lambda_p, \lambda_s + 10\,\mathrm{nm})$ and $\phi(\lambda_p, \lambda_s - 10\,\mathrm{nm})$ which are 10 nm above and 10 nm below the central signal wavelength λ_s are calculated. The difference of these values is shown in the plot

```
In[27]:= (* compensation crystal length, Δλs = 20 nm *)
        Plot[
         180/π (ϕdc[θp, λp, λs - .01, L, LC, Lpc, TBBO, Tc, Tpc] -
             ϕdc[θp, λp, λs + .01, L, LC, Lpc, TBBO, Tc, Tpc]) /.
          sourceparameters, {LC, 3070, 3170},
         FrameLabel→{"LC in μm", "Δϕ in degrees"}, ImageSize→400,
         PlotRange→All]
```

Out[27]=



Even for a 50 μm thinner or thicker compensation crystal the range of the phases is still below 10 degrees.

4.7.3 Pump Wavelength

The phase between $|HH\rangle$ and $|VV\rangle$ depends on the pump wavelength. The width of the spectrum of typical laser diodes is below 1 nm. Results for $L_{\mathrm{pc}} = 3440\,\mu\mathrm{m}$:

```
In[28]:= (* change of λp *)
  Plot[Mod[180 / π (ϕdc[θp, λP, λs, L, Lc, Lpc, TBBO, Tc, Tpc]) /.
      sourceparameters, 360], {λP, .4055, .4065},
    FrameLabel → {"λ in μm", "phase in degrees"}, ImageSize → 400,
    PlotRange → All]
```

Out[28]=



The change over the whole spectrum is smaller than 8 degrees.

4.7.4 Pre-Compensation Crystal Length

The influence of the pre-compensation crystal length `LPC` on the phase compensation is calculated. The phases $\phi(\lambda_p + 0.5\,\mathrm{nm}, \lambda_s)$ and $\phi(\lambda_p - 0.5\,\mathrm{nm}, \lambda_s)$ which are 0.5 nm above and 0.5 nm below the central pump wavelength λ_p are calculated. The difference of these values is shown in the plot

```
In[29]:= (* pre-compensation crystal length, Δλp = 1 nm *)
  Plot[
    180 / π (ϕdc[θp, λp - 0.0005, λs, L, Lc, LPC, TBBO, Tc, Tpc] -
        ϕdc[θp, λp + 0.0005, λs, L, Lc, LPC, TBBO, Tc, Tpc]) /.
     sourceparameters, {LPC, 3390, 3490},
    FrameLabel → {"LPC in μm", "Δϕ in degrees"}, ImageSize → 400,
    PlotRange → All]
```

Out[29]=



The length of the pre-compensation is critical. The length tolerance should be below 10 μm.

4.7.5 BBO Crystal Length

The influence of the BBO crystal length LBBO on the phase compensation is calculated. The phases $\phi(\lambda_p + 0.5\,\mathrm{nm}, \lambda_s)$ and $\phi(\lambda_p - 0.5\,\mathrm{nm}, \lambda_s)$ which are 0.5 nm above and 0.5 nm below the central pump wavelength λ_p are calculated. The difference of these values is shown in the plot

```
In[30]:= (* BBO crystal length, Δλp = 1 nm *)
Plot[
 180/π (ϕdc[Θp, λp - 0.0005, λs, LBBO, Lc, Lpc, TBBO, Tc, Tpc] -
     ϕdc[Θp, λp + 0.0005, λs, LBBO, Lc, Lpc, TBBO, Tc, Tpc]) /.
  sourceparameters, {LBBO, 5950, 6050},
 FrameLabel→{"LBBO in µm", "Δϕ in degrees"}, ImageSize→400,
 PlotRange→All]
```

Out[30]=



4.7.6 Temperature Dependence, Source Without Pre-Compensation

All crystals (BBO,BBO,YVO4) at the same temperature. Temperature dependence of a SPDC source without a pre-compensation crystal:

```
In[92]:= (* Temperature change, no pre-compensation YVO4 *)
Plot[Mod[180/π (ϕdc[Θp, λp, λs, L, Lc, 0*Lpc, T, T, T]) /.
   sourceparameters, 360], {T, 20, 21},
 FrameLabel→{"Temperature [°C]", "phase in degrees"},
 ImageSize→400, PlotRange→All]
```

Out[92]=



If the Temperature of the crystals and the compensation crystal changes by 1 K the phase changes by 40 degrees.

4.7.7 Temperature Dependence, Source With Pre-Compensation

All crystals (YVO4, BBO,BBO,YVO4) at the same temperature. Temperature dependence of a SPDC source with a pre-compensation crystal:

```
In[93]:= (* Temperature change, with pre-compensation YVO4 *)
Plot[Mod[180/π (ϕdc[θp, λp, λs, L, Lc, Lpc, T, T, T]) /.
     sourceparameters, 360], {T, 20, 21},
  FrameLabel→{"Temperature [°C]", "phase in degrees"},
  ImageSize→400, PlotRange→All]
```

Out[93]=



If the temperature of the BBO crystals and the compensation crystal changes by 1 K the phase changes by 8 degrees:

The phase changes in temperature caused by the compensation and pre-compensation crystal have different signs. Therefore the source is less critical in temperature changes if a pre-compensation crystal is used.

4.7.8 Conclusion

The pre-compensation crystal length (relative to the first BBO crystal) and the spectral bandwidth of the pump beam are the most critical parts in phase compensation. The model only uses plane waves in one direction. If long crystals are used the transverse walk-off has to be compensated with additional crystals. If the spatial overlap of $|HH\rangle$ and $|VV\rangle$ photon pairs is not perfect the quantum state is not maximally entangled.

The phase compensation can be tested in several ways. One possibility is to measure the polarization density matrix at different signal (or idler) wavelengths. The phase and the purity of these polarization density matrices provide information on how good the phases are compensated and how good the spatial overlap is.

A different way to test the phase compensation is to measure the wavelength dependence of the polarization rotation of each crystal.

4.8 Single Crystal Polarization Rotation

Wrong length or material properties of the pre-compensation, compensation and BBO crystals have strong influence on the entanglement quality of the two-photon state. The crystals can be tested by measuring the dispersion



Figure 7: Crystal between two crossed polarizers. The polarizers are rotated by ±45 degrees to the ordinary (extraordinary) polarization.

characteristics. One method is to put the crystal between two crossed polarizers at ±45 degrees. The input polarization is

$$|P_{\text{in}}\rangle = \frac{1}{\sqrt{2}}\left(|o\rangle + |e\rangle\right) \tag{46}$$

Behind the crystal the polarization is rotated

$$|P_L\rangle = \frac{1}{\sqrt{2}}\left(|o\rangle + e^{i(k_e-k_o)\cdot L}|e\rangle\right). \tag{47}$$

A wavelength sensitive detector (spectrometer) or a wavelength tunable light source behind the second polarizer at −45 degrees

$$|P_{\text{Det}}\rangle = \frac{1}{\sqrt{2}}\left(|o\rangle - |e\rangle\right) \tag{48}$$

can be used to analyze the rotation. The signal at the detector is

$$I_{\text{Det}} = |\langle P_{\text{Det}}|P_L\rangle|^2 = \sin^2\left(\frac{k_e - k_o}{2}L\right). \tag{49}$$

This signal I_{Det} [2] can be compared to the SPDC phase compensation calculations.

4.8.1 Mathematica Script

The Mathematica script `single_crystal_pol_rotation.nb` can be used to calculate the polarization rotation by a YVO4 or BBO crystal. Linear polarization (direction α) is described by the vector `pol`.
The phase shift between $|o\rangle$ and $|e\rangle$ is represented by the matrix `phasematrix`

phase calculation examples

```
In[22]:= (* polarization state vector *)
pol[α_] := {Sin[α], Cos[α]};
(* phase between {1,0} (o-polarization) and {0,1} (e-polarization)
 vector *)
phasematrix[ϕ_] := {{1, 0}, {0, Exp[-I * ϕ]}};
```

[2]The light beam size should be larger than the beam walk-off. The detector has to be in the overlap region of the ordinary and extraordinary beam.

The signal behind two polarizers is `IDet`. The input and output linear polarization angles are α`in` and α`out`. The functions `IDetBBO` and `IDetYVO` use the dispersion relations of BBO and YVO4. λ is the wavelength in vacuum, `LBBO` and `LYVO` the length of the crystals. `TBBO` and `TYVO` is the temperature of the crystals.

```
In[24]:= IDet[αin_, αout_, ϕ_] := Abs[pol[αout].phasematrix[ϕ].pol[αin]]^2;
IDetBBO[αin_, αout_, θp_, λ_, LBBO_, TBBO_] :=
 IDet[αin, αout, ϕBBO[θp, λ, LBBO, TBBO]];
IDetYVO[αin_, αout_, θ_, λ_, LYVO_, TYVO_] :=
 IDet[αin, αout, ϕYVO[θ, λ, LYVO, TYVO]]
```

The absolute phase between $|o\rangle$ and $|e\rangle$ of the pre-compensation cystal depending on wavelength

```
In[31]:= Plot[180/π (ϕYVO[θYVO, λ, LYVO, TYVO]) /.
   {αin → π/4, αout → -π/4, θYVO → π/2, LYVO → 3440, TYVO → 20},
  {λ, .740, .9}, FrameLabel → {"λ in μm", "phase in degrees"},
  ImageSize → 400, PlotRange → All]
```

Out[31]=



Pre-compensation crystal between two polarizers (±45 degrees). Transmission depends on wavelength:

```
In[27]:= Plot[IDetYVO[αin, αout, θYVO, λ, LYVO, TYVO] /.
   {αin → π/4, αout → -π/4, θYVO → π/2, LYVO → 3440, TYVO → 20},
  {λ, .4, .401}]
```

Out[27]=



5 Polarization State / Density Operator Formalism

The quantum state generated in a SPDC source depends on several degrees of freedom. The polarization state can calculated with the density operator formalism.

A pure photon pair state in a superposition of $|HH\rangle$ and $|VV\rangle$ photons can be written as

$$|\phi\rangle_{\mathrm{P}} = |HH\rangle + e^{i\phi}|VV\rangle. \tag{50}$$

The quantum state of a collinear two crossed crystal SPDC source has a broad spectrum and the quantum state of the photon pairs has to be described by

$$|\phi\rangle = \iint_0^\infty \mathrm{d}\omega_s\,\mathrm{d}\omega_i\,A(\omega_s,\omega_i)|\omega_s,\omega_i\rangle \otimes \left(|HH\rangle_{s,i} + e^{i\varphi(\omega_p,\omega_s,\omega_i)}|VV\rangle_{s,i}\right) \tag{51}$$

where $A(\omega_s,\omega_i)$ is the spectral amplitude. If the phase φ depends on the angular frequencies ω_s, ω_i and ω_p the polarization part of the quantum state can be calculated with the density operator formalism. The density operator is

$$\begin{aligned}
\rho &= |\phi\rangle\langle\phi| \\
&= \left(\iint_0^\infty \mathrm{d}\omega_s\,\mathrm{d}\omega_i\,A(\omega_s,\omega_i)|\omega_s,\omega_i\rangle \otimes \left(|HH\rangle + e^{i\varphi(\omega_p,\omega_s,\omega_i)}|VV\rangle\right)\right) \\
&\qquad \left(\iint_0^\infty \mathrm{d}\tilde{\omega}_s\,\mathrm{d}\tilde{\omega}_i\,\bar{A}(\tilde{\omega}_s,\tilde{\omega}_i)\langle\tilde{\omega}_s,\tilde{\omega}_i| \otimes \left(\langle HH| + e^{-i\varphi(\omega_p,\tilde{\omega}_s,\tilde{\omega}_i)}\langle VV|\right)\right) \\
&= \iint_0^{\omega_p} \mathrm{d}\omega_s\,\mathrm{d}\omega_i \iint_0^{\omega_p} \mathrm{d}\tilde{\omega}_s\,\mathrm{d}\tilde{\omega}_i\,A(\omega_s,\omega_i)\,\bar{A}(\tilde{\omega}_s,\tilde{\omega}_i) \\
&\qquad \left(|HH\rangle + e^{i\varphi(\omega_p,\omega_s,\omega_i)}|VV\rangle\right)\left(\langle HH| + e^{-i\varphi(\omega_p,\tilde{\omega}_s,\tilde{\omega}_i)}\langle VV|\right) \otimes |\omega_s\omega_i\rangle\langle\tilde{\omega}_s,\tilde{\omega}_i|.
\end{aligned} \tag{52}$$

The polarization state is the partial trace over the angular frequencies

$$\begin{aligned}
\rho_p &= \mathrm{Tr}_{\omega_s,\omega_i}\{\rho\} \\
&= \iint_0^\infty \mathrm{d}\omega_s'\,\mathrm{d}\omega_i'\,\langle\omega_s'\omega_i'|\,\rho\,|\omega_s'\omega_i'\rangle \\
&= \iint_0^\infty \mathrm{d}\omega_s'\,\mathrm{d}\omega_i' \iint_0^\infty \mathrm{d}\omega_s\,\mathrm{d}\omega_i \iint_0^\infty \mathrm{d}\tilde\omega_s\,\mathrm{d}\tilde\omega_i\,A(\omega_s,\omega_i)\,\bar A(\tilde\omega_s,\tilde\omega_i) \\
&\qquad \left(|HH\rangle + e^{i\varphi(\omega_p,\omega_s,\omega_i)}|VV\rangle\right)\left(\langle HH| + e^{-i\varphi(\omega_p,\tilde\omega_s,\tilde\omega_i)}\langle VV|\right) \\
&\qquad \delta(\omega_s'-\omega_s)\delta(\omega_i'-\omega_i)\delta(\omega_s'-\tilde\omega_s)\delta(\omega_i'-\tilde\omega_i) \\
&= \iint_0^\infty \mathrm{d}\omega_s'\,\mathrm{d}\omega_i'\,|A(\omega_s',\omega_i')|^2 \\
&\qquad (|HH\rangle\langle HH| + |VV\rangle\langle VV| \\
&\qquad + e^{-i\varphi(\omega_p,\omega_s',\omega_i')}|HH\rangle\langle VV| + e^{i\varphi(\omega_p,\omega_s',\omega_i')}|VV\rangle\langle HH|\Big)
\end{aligned} \tag{53}$$

The quantum state has to be normalized $\mathrm{Tr}\{\rho_P\} = 1$. Therefore

$$\iint_0^\infty \mathrm{d}\omega_s'\,\mathrm{d}\omega_i'\,|A(\omega_s',\omega_i')|^2 = \frac{1}{2} \tag{54}$$

The polarization state is maximally entangled if the absolute value of

$$C_{|HH\rangle\langle VV|} = \iint_0^\infty \mathrm{d}\omega_s'\,\mathrm{d}\omega_i'\,|A(\omega_s',\omega_i')|^2\,e^{-i\varphi(\omega_p,\omega_s',\omega_i')} \tag{55}$$

is $\frac{1}{2}$. If φ is independent of the angular frequencies ω_s, ω_i and ω_p the polarization state is maximally entangled.
C and φ are defined by

$$C \cdot e^{i\varphi} = C_{|HH\rangle\langle VV|}. \tag{56}$$

The polarization density matrix can be written as

$$\rho_p = \frac{1}{2}\left(|HH\rangle\langle HH| + |VV\rangle\langle VV| + C\cdot e^{i\varphi}|HH\rangle\langle VV| + C\cdot e^{-i\varphi}|VV\rangle\langle HH|\right) \tag{57}$$

Linear polarization measurement along the directions α and β

$$\begin{aligned}
|\alpha\rangle &= \sin\alpha|H\rangle + \cos\alpha|V\rangle \\
|\beta\rangle &= \sin\beta|H\rangle + \cos\beta|V\rangle
\end{aligned} \tag{58}$$

of the signal and idler photon

$$\begin{aligned}
|\alpha\beta\rangle = |\alpha\rangle|\beta\rangle &= \sin\alpha\cdot sin\beta|HH\rangle + \cos\alpha\cdot\cos\beta|VV\rangle \\
&\quad + \sin\alpha\cdot\cos\beta|HV\rangle + \cos\alpha\cdot\sin\beta|VH\rangle
\end{aligned} \tag{59}$$

result in a detection probability

$$
\begin{aligned}
P_{\alpha,\beta} = \langle \alpha\beta | \rho_P | \alpha\beta \rangle &= \frac{1}{2}(\sin^2\alpha \cdot \sin^2\beta + \cos^2\alpha \cdot \cos^2\beta \\
&\qquad + 2C\cos\varphi \cdot \sin\alpha \cdot \cos\alpha \cdot \sin\beta \cdot \cos\beta) \\
&= \frac{1}{4}(1 + \cos(2\alpha) \cdot \cos(2\beta) + C\cos\varphi \cdot \sin(2\alpha) \cdot \sin(2\beta))
\end{aligned}
\tag{60}
$$

The visibility is defined by

$$
\text{vis} \equiv \frac{P_{\max} - P_{\min}}{P_{\max} + P_{\min}} \tag{61}
$$

Using $P_{\alpha,\beta}$ it is

$$
\begin{aligned}
\text{vis}_{H/V} &= 1 \\
\text{vis}_{\pm 45} &= |C\cos\varphi|
\end{aligned}
\tag{62}
$$

If only one polarizer is used and the correlation behind this polarizer is measured the angle α is the same as $\beta = \alpha$. And

$$
P_{\alpha=\beta} = \frac{1}{2}(1 + \cos^2(2\alpha) + C\cos\varphi \cdot \sin^2(2\alpha)). \tag{63}
$$

The visibility of this measurement is

$$
\text{vis}_{\alpha=\beta} = \frac{2}{3 + C \cdot \cos\varphi} \tag{64}
$$

Gaussian Angular Frequency Spectrum and Linear Phase Model

A simple model can be used to calculate the polarization density matrix:

- central angular frequency is ω_s
- Gaussian distribution around ω_s
- perfect phase matching $\omega_p = \omega_s + \omega_i$
- phase between $|HH\rangle$ and $|VV\rangle$ depends linear on ω_s ,ω_i and ω_p

The amplitude of the two-photon state is

$$
|A(\omega_s,\omega_i)|^2 = A_0 e^{-(\omega_c - \omega_s)^2/\Delta_s^2} \cdot \delta(\omega_p - \omega_s - \omega_i) \tag{65}
$$

The phase is the sum of signal, idler and pump phase

$$
\begin{aligned}
\varphi(\omega_p,\omega_s,\omega_i) &= \varphi_p(\omega_p) + \varphi_s(\omega_s) + \varphi_i(\omega_i) \\
&= \varphi_0 + \varphi_{p0} \cdot \omega_p + \varphi_{s0} \cdot \omega_s + \varphi_{i0} \cdot \omega_i
\end{aligned}
\tag{66}
$$

The constants φ_{s0}, φ_{i0} and φ_{p0} only depend on the used crystals. The integral (eqn. 55) is

$$
\begin{aligned}
C_{|HH\rangle\langle VV|} &= \iint_0^\infty \mathrm{d}\omega_s' \, \mathrm{d}\omega_i' \, |A(\omega_s',\omega_i')|^2 \, e^{-i\varphi(\omega_p,\omega_s',\omega_i')} \\
&= \iint_0^\infty \mathrm{d}\omega_s' \, \mathrm{d}\omega_i' \, A_0 e^{-(\omega_c-\omega_s')^2/\Delta_s^2} \cdot \delta(\omega_p - \omega_s' - \omega_i') e^{-i(\varphi_0+\varphi_{p0}\cdot\omega_p+\varphi_{s0}\cdot\omega_s'+\varphi_{i0}\cdot\omega_i')} \\
&= A_0 e^{-i\varphi_0 - i(\varphi_{p0}+\varphi_{i0})\cdot\omega_p} \int_0^\infty \mathrm{d}\omega_s' \, e^{-(\omega_c-\omega_s')^2/\Delta_s^2} \cdot e^{-i(\varphi_{s0}-\varphi_{i0})\cdot\omega_s'} \\
&= A_0 e^{-i\varphi_0 - i(\varphi_{p0}+\varphi_{i0})\cdot\omega_p} e^{-i(\varphi_{s0}-\varphi_{i0})\omega_c} \Delta\sqrt{\pi} e^{-\frac{1}{4}(\varphi_{s0}-\varphi_{i0})\Delta_s^2}
\end{aligned}
\tag{67}
$$

and the absolute value

$$
\left|C_{|HH\rangle\langle VV|}\right|^2 = \frac{1}{2} \, e^{-\frac{1}{4}(\varphi_{s0}-\varphi_{i0})\Delta_s^2} \tag{68}
$$

The values φ_{s0} and φ_{i0} depend on the used crystals.

6 Pump Beam Walk-Off



Figure 8: walk-off effect of ordinary and extraordinary beams in birefringent crystals.

walk off angle ρ in uniaxial crystals [2] (eqn. 69, page 98)

$$
\tan\rho = \frac{1}{2}\left[n_e(\theta)\right]^2 \left[\frac{1}{n_e^2} - \frac{1}{n_o^2}\right] \sin(2\theta) \tag{69}
$$

with

$$
\frac{1}{\left[n_e(\theta)\right]^2} = \frac{\cos^2(\theta)}{n_o^2} + \frac{\sin^2(\theta)}{n_e^2} \tag{70}
$$

[2] (eqn. 48, page 58) principal refractive indices $n_x = n_y = n_o$ and $n_z = n_e$. θ angle to z-axis.

References

[1] Nicolas Boeuf, David Branning, Ivan Chaperot, Eric Dauler, S Gue, Gregg Jaeger, Antoine Muller, Alan Migdall, et al. Calculating characteristics of noncollinear phase matching in uniaxial and biaxial crystals. *Optical Engineering*, 39(4):1016–1024, 2000.

[2] Richard L Sutherland. *Handbook of nonlinear optics.* CRC press, 2003.

[3] Pavel Trojek. *Efficient generation of photonic entanglement and multiparty quantum communication.* PhD thesis, lmu, 2007.

[4] Pavel Trojek and Harald Weinfurter. Collinear source of polarization-entangled photon pairs at nondegenerate wavelengths. *Applied Physics Letters*, 92(21):211103, 2008.